tions. Also, you might sometimes want to batch-fit to a number of data sets without interacting with the dialog for each data set. These circumstances will require that you use command lines to do fits.

The easiest way to do this is to use the Curve Fitting dialog to generate command lines that do almost what you want. If you simply want to add a more complex feature, such as a more complicated constraint expression, click the To Cmd Line button, then edit the commands generated by the dialog to add the features you want.

If you are writing a user procedure that does curve fitting, you can click the To Clip button to copy the commands generated by the dialog. Then paste the commands into the Procedure window. Edit them as needed by your application.

Curve fitting is done by three operations — **CurveFit**, **FuncFit**, and **FuncFitMD**. You will find details on these operations in Chapter V-1, **Igor Reference**.

Batch Fitting

If you are doing batch fitting, you probably want to call a curve fitting operation inside a loop. In that case, you probably don't want a history report for every fit- it could possible make a very large amount of text in the history. You probably don't want the progress window during the fits- it slows down the fit and will make a flashing window as it appears and disappears. Finally, you probably don't want graphs and tables updating during the fits, as this can slow down computation considerably.

Here is an example function that will do all of this, plus it checks for an error during the fit. If the use of the $ operator is unfamiliar you will want to consult **Accessing Waves in Functions** on page IV-82**.**

```
Function FitExpToListOfWaves(theList)
   String theList

   Variable i=0
   string aWaveName = ""
   Variable V_fitOptions = 4     // suppress progress window
   Variable V_FitError = 0       // prevent abort on error
   do
      aWaveName = StringFromList(i, theList)
      WAVE/Z aWave = $aWaveName
      if (!WaveExists(aWave))
         break
      endif

      // /N suppresses screen updates during fitting
      // /Q suppresses history output during fitting
      CurveFit/N/Q exp aWave /D/R
      WAVE W_coef

      // save the coefficients
      Duplicate/O W_coef $("cf_"+aWaveName)
      // save errors
      Duplicate/O W_sigma, $("sig_"+aWaveName)
      if (V_FitError != 0)
         // Mark the results as being bad
         WAVE w = $("cf_"+aWaveName)
         w = NaN
         WAVE w = $("sig_"+aWaveName)
         w = NaN
         WAVE w = $("fit_"+aWaveName)
         w = NaN
         WAVE w = $("Res_"+aWaveName)
         w = NaN
         V_FitError = 0
      endif
      i += 1
```